Phonon Dispersion of a sc lattice

Paul Wenk, 01.05.18

Connections to nearest neighbors (nn)

In[280]:=
```
Δ = Union[Permutations[{a, 0, 0}], Permutations[{-a, 0, 0}]]
```

Out[280]=
```
{{0, 0, -a}, {0, 0, a}, {0, -a, 0}, {0, a, 0}, {-a, 0, 0}, {a, 0, 0}}
```

Connections to next nearest neighbors (nnn)

In[281]:=
```
ΔP = Union[Permutations[{a, a, 0}],
  Permutations[{-a, -a, 0}], Permutations[{-a, a, 0}]]
```

Out[281]=
```
{{0, -a, -a}, {0, -a, a}, {0, a, -a}, {0, a, a}, {-a, 0, -a}, {-a, 0, a},
 {-a, -a, 0}, {-a, a, 0}, {a, 0, -a}, {a, 0, a}, {a, -a, 0}, {a, a, 0}}
```

The contribution from nn :

We sum over all possible nn neighbors (index i) and vector components (dim1={x,y,z}, dim2={x,y,z} with x≡1, y≡2, z≡3). The u[] functions are the small elongations.

In[284]:=
```
Clear[C1, i, dim1, dim2, u, n, m, l, a];
Sum[
   C1/a^2 Δ[[i]][[dim1]] Δ[[i]][[dim2]] (u[{n, m, l}, dim1] u[{n, m, l}, dim2] -
      u[{n, m, l}, dim1] u[{n, m, l} + Δ[[i]]/a, dim2]),
   {i, 1, Length[Δ]}, {dim1, 1, 3}, {dim2, 1, 3}] // FullSimplify // Expand
```

Out[285]=
```
-C1 u[{-1 + n, m, l}, 1] u[{n, m, l}, 1] +
 2 C1 u[{n, m, l}, 1]^2 - C1 u[{n, -1 + m, l}, 2] u[{n, m, l}, 2] +
 2 C1 u[{n, m, l}, 2]^2 - C1 u[{n, m, -1 + l}, 3] u[{n, m, l}, 3] +
 2 C1 u[{n, m, l}, 3]^2 - C1 u[{n, m, l}, 3] u[{n, m, 1 + l}, 3] -
 C1 u[{n, m, l}, 2] u[{n, 1 + m, l}, 2] - C1 u[{n, m, l}, 1] u[{1 + n, m, l}, 1]
```

To get only the xx component, we set dim1=1, dim2=1

In[286]:=

```
S1 = Sum[
  C1/a^2 Δ[[i]][[dim1]] Δ[[i]][[dim2]] (u[{n, m, l}, dim1] u[{n, m, l}, dim2] -
     u[{n, m, l}, dim1] u[{n, m, l} + Δ[[i]]/a, dim2]),
  {i, 1, Length[Δ]}, {dim1, 1}, {dim2, 1}] // FullSimplify // Expand
```

Out[286]=

```
-C1 u[{-1 + n, m, l}, 1] u[{n, m, l}, 1] +
 2 C1 u[{n, m, l}, 1]^2 - C1 u[{n, m, l}, 1] u[{1 + n, m, l}, 1]
```

The contribution from nnn:

In[287]:=

```
Sum[
  C2/(2 a^2) ΔP[[i]][[dim1]] ΔP[[i]][[dim2]] (u[{n, m, l}, dim1] u[{n, m, l}, dim2] -
     u[{n, m, l}, dim1] u[{n, m, l} + ΔP[[i]]/a, dim2]),
  {i, 1, Length[ΔP]}, {dim1, 1, 3}, {dim2, 1, 3}] // FullSimplify // Expand
```

Out[287]=

```
-1/2 C2 u[{-1 + n, -1 + m, l}, 1] u[{n, m, l}, 1] -
 1/2 C2 u[{-1 + n, -1 + m, l}, 2] u[{n, m, l}, 1] -
 1/2 C2 u[{-1 + n, m, -1 + l}, 1] u[{n, m, l}, 1] -
 1/2 C2 u[{-1 + n, m, -1 + l}, 3] u[{n, m, l}, 1] -
 1/2 C2 u[{-1 + n, m, 1 + l}, 1] u[{n, m, l}, 1] +
 1/2 C2 u[{-1 + n, m, 1 + l}, 3] u[{n, m, l}, 1] -
 1/2 C2 u[{-1 + n, 1 + m, l}, 1] u[{n, m, l}, 1] +
 1/2 C2 u[{-1 + n, 1 + m, l}, 2] u[{n, m, l}, 1] + 4 C2 u[{n, m, l}, 1]^2 -
 1/2 C2 u[{-1 + n, -1 + m, l}, 1] u[{n, m, l}, 2] -
 1/2 C2 u[{-1 + n, -1 + m, l}, 2] u[{n, m, l}, 2] +
 1/2 C2 u[{-1 + n, 1 + m, l}, 1] u[{n, m, l}, 2] -
 1/2 C2 u[{-1 + n, 1 + m, l}, 2] u[{n, m, l}, 2] -
 1/2 C2 u[{n, -1 + m, -1 + l}, 2] u[{n, m, l}, 2] -
```

```
1/2 C2 u[{n, -1 + m, -1 + l}, 3] u[{n, m, l}, 2] -
1/2 C2 u[{n, -1 + m, 1 + l}, 2] u[{n, m, l}, 2] +
1/2 C2 u[{n, -1 + m, 1 + l}, 3] u[{n, m, l}, 2] + 4 C2 u[{n, m, l}, 2]^2 -
1/2 C2 u[{-1 + n, m, -1 + l}, 1] u[{n, m, l}, 3] -
1/2 C2 u[{-1 + n, m, -1 + l}, 3] u[{n, m, l}, 3] +
1/2 C2 u[{-1 + n, m, 1 + l}, 1] u[{n, m, l}, 3] -
1/2 C2 u[{-1 + n, m, 1 + l}, 3] u[{n, m, l}, 3] -
1/2 C2 u[{n, -1 + m, -1 + l}, 2] u[{n, m, l}, 3] -
1/2 C2 u[{n, -1 + m, -1 + l}, 3] u[{n, m, l}, 3] +
1/2 C2 u[{n, -1 + m, 1 + l}, 2] u[{n, m, l}, 3] -
1/2 C2 u[{n, -1 + m, 1 + l}, 3] u[{n, m, l}, 3] +
4 C2 u[{n, m, l}, 3]^2 - 1/2 C2 u[{n, m, l}, 2] u[{n, 1 + m, -1 + l}, 2] +
1/2 C2 u[{n, m, l}, 3] u[{n, 1 + m, -1 + l}, 2] +
1/2 C2 u[{n, m, l}, 2] u[{n, 1 + m, -1 + l}, 3] -
1/2 C2 u[{n, m, l}, 3] u[{n, 1 + m, -1 + l}, 3] -
1/2 C2 u[{n, m, l}, 2] u[{n, 1 + m, 1 + l}, 2] - 1/2 C2 u[{n, m, l}, 3] u[{n, 1 + m, 1 + l}, 2] -
1/2 C2 u[{n, m, l}, 2] u[{n, 1 + m, 1 + l}, 3] - 1/2 C2 u[{n, m, l}, 3] u[{n, 1 + m, 1 + l}, 3] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, -1 + m, l}, 1] +
1/2 C2 u[{n, m, l}, 2] u[{1 + n, -1 + m, l}, 1] +
1/2 C2 u[{n, m, l}, 1] u[{1 + n, -1 + m, l}, 2] -
1/2 C2 u[{n, m, l}, 2] u[{1 + n, -1 + m, l}, 2] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, -1 + l}, 1] +
1/2 C2 u[{n, m, l}, 3] u[{1 + n, m, -1 + l}, 1] +
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, -1 + l}, 3] -
```

```
1/2 C2 u[{n, m, l}, 3] u[{1 + n, m, -1 + l}, 3] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, 1 + l}, 1] - 1/2 C2 u[{n, m, l}, 3] u[{1 + n, m, 1 + l}, 1] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, 1 + l}, 3] - 1/2 C2 u[{n, m, l}, 3] u[{1 + n, m, 1 + l}, 3] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, 1 + m, l}, 1] - 1/2 C2 u[{n, m, l}, 2] u[{1 + n, 1 + m, l}, 1] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, 1 + m, l}, 2] - 1/2 C2 u[{n, m, l}, 2] u[{1 + n, 1 + m, l}, 2]
```

To get ony the xx component, we set dim1=1,dim2=1

In[288]:=

```
S2 = Sum[
   C2/(2 a^2) ΔP[[i]][[dim1]] ΔP[[i]][[dim2]] (u[{n, m, l}, dim1] u[{n, m, l}, dim2] -
      u[{n, m, l}, dim1] u[{n, m, l} + ΔP[[i]]/a, dim2]),
   {i, 1, Length[ΔP]}, {dim1, 1}, {dim2, 1}] // FullSimplify // Expand
```

Out[288]=

```
-1/2 C2 u[{-1 + n, -1 + m, l}, 1] u[{n, m, l}, 1] -
1/2 C2 u[{-1 + n, m, -1 + l}, 1] u[{n, m, l}, 1] -
1/2 C2 u[{-1 + n, m, 1 + l}, 1] u[{n, m, l}, 1] -
1/2 C2 u[{-1 + n, 1 + m, l}, 1] u[{n, m, l}, 1] +
4 C2 u[{n, m, l}, 1]^2 - 1/2 C2 u[{n, m, l}, 1] u[{1 + n, -1 + m, l}, 1] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, -1 + l}, 1] -
1/2 C2 u[{n, m, l}, 1] u[{1 + n, m, 1 + l}, 1] - 1/2 C2 u[{n, m, l}, 1] u[{1 + n, 1 + m, l}, 1]
```

Inserting the ansatz to get an algebraical equation

In[289]:=

```
Clear[u]
```

In[290]:=

```
u[a_, b_] := Evaluate[Exp[I a.{qx, qy, qz}]] /. {n → 0, m → 0, l → 0};
```

In[291]:=

```
S1 /. u[{n, m, l}, 1] → 1 // FullSimplify
```

Out[291]=

```
-2 C1 (-1 + Cos[qx])
```

In[292]:=
```
S2 /. u[{n, m, l}, 1] → 1 // FullSimplify
```
Out[292]=
```
-2 C2 (-2 + Cos[qx] (Cos[qy] + Cos[qz]))
```

The xy component

In[293]:=
```
dim1 = 1;
dim2 = 2;
S3 = Sum[
    C2/(2 a^2) ΔP[[i]][[dim1]] ΔP[[i]][[dim2]] (u[{n, m, l}, dim1] u[{n, m, l}, dim2] -
        u[{n, m, l}, dim1] u[{n, m, l} + ΔP[[i]]/a, dim2]),
    {i, 1, Length[ΔP]}] // FullSimplify // Expand
Clear[dim1, dim2]
```
Out[295]=
```
2 C2 Sin[qx] Sin[qy]
```

Inserting the ansatz

In[297]:=
```
S3 /. u[{n, m, l}, 1] → 1 // FullSimplify
```
Out[297]=
```
2 C2 Sin[qx] Sin[qy]
```

The dynamical matrix

Applying cyclic permutations in x,y,z to the pervious results, we get all components of the dynamical matrix Dyn

here we set M=1

In[251]:=
```
Dyn[{kx, ky, kz}, C1, C2]
```
Out[251]=
```
{{2 C1 (1 - Cos[kx]) + 2 C2 (2 - Cos[kx] Cos[ky] - Cos[kx] Cos[kz]),
  2 C2 Sin[kx] Sin[ky], 2 C2 Sin[kx] Sin[kz]},
 {2 C2 Sin[kx] Sin[ky], 2 C1 (1 - Cos[ky]) + 2 C2 (2 - Cos[kx] Cos[ky] - Cos[ky] Cos[kz]),
  2 C2 Sin[ky] Sin[kz]}, {2 C2 Sin[kx] Sin[kz], 2 C2 Sin[ky] Sin[kz],
  2 C1 (1 - Cos[kz]) + 2 C2 (2 - Cos[kx] Cos[kz] - Cos[ky] Cos[kz])}}
```

For **k**={k,0,0} we get the following einegvalues

In[253]:=
```
Eigenvalues[Dyn[{k, 0, 0}, C1, C2]] // MatrixForm
```
Out[253]//MatrixForm=
```
( (-C1 - 2 C2) (-2 + 2 Cos[k]) )
(     -C2 (-2 + 2 Cos[k])      )
(     -C2 (-2 + 2 Cos[k])      )
```

Plotting the spectrum

In[271]:=

```
Manipulate[
 Dyn[k_, C1_, C2_] :=
  {{2 C1 (1 - Cos[k[[1]]]) + 2 C2 (2 - Cos[k[[1]]] Cos[k[[2]]] - Cos[k[[1]]] Cos[k[[3]]]),
    2 C2 Sin[k[[1]]] Sin[k[[2]]], 2 C2 Sin[k[[1]]] Sin[k[[3]]]}, {2 C2 Sin[k[[1]]] Sin[k[[2]]],
    2 C1 (1 - Cos[k[[2]]]) + 2 C2 (2 - Cos[k[[1]]] Cos[k[[2]]] - Cos[k[[2]]] Cos[k[[3]]]),
    2 C2 Sin[k[[2]]] Sin[k[[3]]]}, {2 C2 Sin[k[[1]]] Sin[k[[3]]], 2 C2 Sin[k[[2]]] Sin[k[[3]]],
    2 C1 (1 - Cos[k[[3]]]) + 2 C2 (2 - Cos[k[[1]]] Cos[k[[3]]] - Cos[k[[2]]] Cos[k[[3]]])}};
 EigenCalc[kmax0_, θ0_, ϕ0_, C10_, C20_] :=
  Module[{tab, kmax = kmax0, θ = θ0, ϕ = ϕ0, C1 = C10, C2 = C20},
   tab = Flatten[Table[{k, √#} & /@ Eigenvalues[Dyn[Pi k {Sin[θ] Cos[ϕ],
          Sin[θ] Sin[ϕ], Cos[θ]}, C1, C2]], {k, 0, kmax, kmax/200}], 1];
   tab];
 tab = EigenCalc[kmax, θ, ϕ, C1, C2];
 Column[{
   ListPlot[tab, PlotRange → All, Frame → True, FrameLabel → {"k/π", "ω(k)"},
    PlotLabel → "Phonon Dispersion", ImageSize → Medium],
   Grid[{
     {"kmax", kmax},
     {"C1", C1},
     {"C2", C2},
     {"θ/π", θ / π},
     {"ϕ/π", ϕ / π}
    }]}],
 {C1, 1, 10}, {C2, 1, 10}, {kmax, 0.001, 3}, {θ, 0, π}, {ϕ, 0, 2 π}
]
```

Out[271]=


C1
C2
kmax
θ
ϕ
Phonon Dispersion
ω(k)
k/π
0
2
4
6
8
0.0
0.5
1.0
1.5
2.0
2.5
3.0
kmax 3.
C1 1.
C2 10.
θ/π 0.63
ϕ/π 0.536

In[272]:=

C1

C2

kmax

θ

0.7891

ϕ



k_{max}	3.
C_1	1.
C_2	10.
θ/π	0.251178
ϕ/π	0.

In[274]:=

C1

C2

kmax

θ

0.785398

ϕ

0.785398



k_{max}	3.
C_1	1.
C_2	10.
θ/π	0.25
ϕ/π	0.25

Out[274]=


C1
C2
kmax
θ
ϕ
Phonon Dispersion
ω(k)
0
2
4
6
8
0.0
0.5
1.0
1.5
2.0
2.5
3.0
k/π
kmax 3.
C1 1.
C2 10.
θ/π 0.25
ϕ/π 0.